Triple Integration with Maple

Maple is helpful in setting up triple iterated integrals over regions in R^3. Just as for double integrals, the key step is determining the limits of integration. That is easier if you imagine an arrow shooting through the region *E* of integration parallel to the axis of the variable with respect to which the first integration takes place. Plotting the region *E* reduces the vividness of imagination necessary to see the corresponding limits.

Fubini's theorem for triple integrals states that the value of a triple integral of a continuous function *f* over a region *E* in R^3 is a triple iterated integral. Suppose that *E* is a "Type 1" region between surfaces with equations $z = h(x, y)$ and $z = k(x, y)$ and has perpendicular projection *D* on the *xy*-plane. Then

$$\int\int\int_E f(x, y, z)\, dz\, dy\, dx = \int\int_D \int_{h(x, y)}^{k(x, y)} f(x, y, z)\, dz\, dy\, dx$$

The remaining double integral over the region *D* in the *xy*-plane is evaluated by iteration as before.

Example 1. Evaluate $\int\int\int_E z\, dz\, dy\, dx$, where *E* is the region between the paraboloid $z = x^2 + y^2$ and the plane $z = 2$.

Solution. The following Maple routine plots the region *E* with an upward-pointing arrow passing through it. The entry point of the arrow occurs on the surface whose equation $z = h(x, y)$ gives the lower limit of the first integration, which will take place with respect to *z*. The exit point of the arrow is on the surface whose equation $z = k(x, y)$ gives the upper limit of that first integration.

As for two-dimensional arrows, when working with other regions modify the parameters in the `arrow` command to size the arrow appropriately. The first and second triples within the command specify the starting and end point of the arrow; the third one specifies a normal direction for the plane in which the arrow is seen: note that it does not appear perfectly vertical even though the direction vector **k** was specified. Adjust the direction vector to get an image that you like.

```
> with(plots):
  with (plottools):
  regionbot := plot3d( x^2 + y^2, x = -2..2, y = -2..2, color = red ):
  regiontop := plot3d( 2, x = -2..2, y = -2..2, color = cyan ):
  regname := textplot3d([0.5, 0.5, 2.5, "E"], font = [TIMES, ITALIC, 14], color = black):
  slice := arrow([.3, .3, -1], [.3, .3, 8.5], [0, 0, 1], 1, 4, .25, color = blue):
  display(regionbot, regiontop, slice, regname, axes = boxed);
Warning, the name changecoords has been redefined
```



Maple's real-time rotation feature reveals that the bottom boundary of the region is the paraboloid $z = x^2 + y^2$, and its top boundary is the plane $z = 2$. The lower limit of the first integration, with respect to *z*, corresponds to the equation of the lower surface, and the upper limit is $z = 2$. Hence,

$$\int\int\int_E \mathrm{f}(x, y, z)\, dz\, dy\, dx = \int\int_{\mathrm{D}} \int_{x^2 + y^2}^{2} z\, dz\, dx\, dy$$

$$= \frac{1}{2} \int\int_D 4 - (x^2 + y^2)^2 \, dx \, dy.$$

As the three-dimensional figure indicates, the region *D* is the disk centered at the origin with radius $\sqrt{2}$. Its equation comes from finding the intersection of $z = x^2 + y^2$ and $z = 2$: equate the two expressions for *z*:

$$x^2 + y^2 = 2.$$

The next routine plots *D* in the *xy*-plane.

```
> with (plots):
  planereg := polarplot( sqrt(2), theta = 0..2*Pi, scaling = constrained ):
  regname := textplot([.7, .7, "D"], font = [TIMES, ITALIC, 14], color = magenta):
  display(planereg, regname);
```

Since the region *D* is a circular disk with boundary $x^2 + y^2 = 2$, change to polor coordinates:

$$\left[\frac{1}{2}\right] \int_0^{2\pi} \int_0^{\sqrt{2}} (4 - r^4) \, r \, dr \, d\theta = \pi \left(4 - \frac{8}{6}\right) = \frac{8\pi}{3}.$$

To check this calculation, have Maple's `student` package evaluate the triple integral over the region *E* by iteration:

```
> with (student):
  value( Tripleint(z, z = x^2 + y^2..2, y = -sqrt(2 - x^2)..sqrt(2 - x^2), x = -sqrt(2)..sqrt(2)) );
```

$$\frac{8}{3}\pi$$

The next example comes from Exercise 10 of Section 16.7, in James Stewart, *Calculus*, *4th Edition*, ITP Brooks/Cole, 1999.

Example 2. Evaluate $\int\int\int_E x \, dz \, dy \, dx$, where is *E* the solid bounded by the planes $x = 0$ (i.e, the *yz*-plane), $y = 0$ (i.e., the *xz*-plane) and $3x + 2y + z = 6$.

Solution. The following modification of the first routine above draws a picture of the region *E* with an upward-pointing arrow through it.

```
> with(plots):
  with (plottools):
  top := plot3d( 6 - 3*x -2*y, x = 0..2, y = 0..3, color = rust ):
  bot := plot3d( 0, x = 0..2, y = 0..3, color = cyan ):
  regname := textplot3d([0.2, 0.2, 2, "E"], font = [TIMES, ITALIC, 14], color = black):
  slice := arrow([.3, .3, -2], [.3, .3, 10], [0, 0, 1], 1, 4, .25, color = blue):
  display(top, bot, slice, regname, axes = boxed, labels = [x, y, z]);
```



This figure and the description of the region *E* show that an arrow shot left-to-right through it enters through the *xz*-plane and exits from the plane $3x + 2y + z = 6$. Thus,

$$\int\int\int_E x\,dz\,dy\,dz = \int\int_D\int_0^{6-3x-2y} x\,dz\,dy\,dx = \int\int_D [6x - 3x^2 - 2xy]\,dy\,dx,$$

where *D* is the triangular region in the *xy*-plane with vertices at the origin, (2, 0) and (0, 3). The following routine plots *D*.

```
> with(plots):
  with (plottools):
  region := plot( 3*(1 - x/2), x = 0..2, color = red ):
  regname := textplot([0.6, 1.3, "D"], font = [TIMES, ITALIC, 14], color = black):
  slice := arrow([.9, -.3], [.9, 2.25], .01, .08, .08, color = blue):
  display(slice, region, regname);
```



Double integration by iteration then gives

$$\int\int_D [6x - 3x^2 - 2xy]\,dy\,dx = \int_0^2\int_0^{3-\frac{3x}{2}} [6x - 3x^2 - 2xy]\,dy\,dx$$

$$= \int_0^2 \left[9x - 9x^2 + \frac{9x^3}{4}\right] dx = 3.$$

The omitted algebra in the last two steps is formidable enough to have Maple's `TripleInt` command check the answer.

```
> with (student):
  value( Tripleint(x, z = 0..6 - 3*x - 2*y, y = 0..3 - 3*x/2, x = 0..2) );
```

3

Note that Maple can't catch an error in the limits of integration, so take care in setting those up!

Thus far, it has been natural to "integrate out" *z* first. That reduces evaluation of triple integrals to *xy*-double integrals. However, Fubini's theorem explicitly states that you can iterate in *any* convenient order: all methods of iteration lead to the same final answer — the value of the triple integral of the continuous function *f* over the region *E*. This worksheet's final example, which is Exercise 18, Section 16.7, of Stewart's *Calculus*, illustrates that.

Example 3. Find the volume of the solid *E* bounded by the elliptical cylinder $4x^2 + z^2 = 4$ and the planes $y = 0$ (the *xz*-plane) and $y = z + 2$.

Solution. It is challenging to draw a good picture of *E* by hand, but not for Maple. To plot the *xz*-plane, whose equation $y = 0$ is not of the form $z = f(x, y)$, the following routine uses parametric plotting. The *xz*-plane is describable by the parametric equations $x = u$, $y = 0$, $z = v$ as *u* and *v* vary over **R**. Note also that the routine omits some of the cylinder to provide a look inside the region.

```
> with(plots):
  with (plottools):
  cyltop := plot3d( sqrt(4 - 4*x^2), x = -1..0.5, y = -2..4,  color = rust ):
  cylbot := plot3d( -sqrt(4 - 4*x^2), x = -1..0.75, y = -2..4, color = rust ):
  plane := plot3d( y - 2, x = -2..2, y = -2..4, color = cyan ):
  coordplane := plot3d( [u, 0, v], u = -2..2, v = -2..2, color = cyan ):
  regname := textplot3d([0.5, 1.5, 1, "E"], font = [TIMES, ITALIC, 14], color = black):
  slice := arrow([.3, -1, .3], [.3, 7, .3], [0, 1, .4], 1, 4, .2, color = blue):
  display(cyltop, cylbot, plane, coordplane, slice, regname, axes = boxed, labels = [x, y, z]);
```



This figure shows that *E* lies inside the elliptic cylinder above the plane $y = z + 2$. Integrating first with respect to *y* gives

$$V = \int\int\int_E 1\, dy\, dz\, dx = \int\int_D \int_0^{z+2} 1\, dy\, dz\, dx,$$

where D is the elliptical region inside $x^2 + \frac{z^2}{4} = 1$, which results from projection of the cylinder onto the *xz*-plane. The next routine draws *D*.

```
> with(plots):
  with (plottools):
  region := implicitplot( x^2 + z^2/4 = 1, x = -1..2, z = -2..2, color = red, scaling = constrained ):
  regname := textplot([0.6, 1.3, "D"], font = [TIMES, ITALIC, 14], color = black):
  slice := arrow([-.5, -2.25], [-.5, 2.25], .025, .15, .08, color = blue):
  display(slice, region, regname, labels = [x, z]);
```



Integration of the above triple integral for *V* with respect to y thus gives

$$V = \int\int_D [z + 2]\, dz\, dx,$$

hand evaluation of which is not appealing: it requires trignometric substitution since term(s) involving $\sqrt{4 - 4x^2}$ or $\sqrt{1 - \frac{z^2}{4}}$ must be antidifferentiated. It thus seems preferable to have Maple complete the problem:

```
> with (student):
  value( Doubleint(z+2, z = -sqrt(4 - 4*x^2)..sqrt(4 - 4*x^2), x = -1..1) );
>
```

$$4\pi$$

As a check on the first integration, with respect to *y*, have Maple evaluate the original triple integral:

```
> with (student):
  value( Tripleint(1, y = 0..z+2, z = -sqrt(4 - 4*x^2)..sqrt(4 - 4*x^2), x = -1..1) );
```

$$4\pi$$